



*International*

*Virtual*

*Observatory*

*Alliance*

# IVOA Astronomical Data Query Language

## Version 1.5

*IVOA Working Draft 2007 September 3*

**This version:**

1.5-20070903

**Latest version:**

<http://www.ivoa.net/Documents/latest/ADQL.html>

**Previous version(s):**

none

**Editor(s):**

Pedro Osuna and Inaki Ortiz

**Author(s):**

Inaki Ortiz, Jeff Lusted, Alexander Szalay, Yuji Shirasaki, Maria A. Nieto-Santisteban, Masatoshi Ohishi, William O'Mullane, Pedro Osuna, the VOQL-TEG and the VOQL Working Group.

---

## Abstract

This document describes the Astronomical Data Query Language (ADQL). ADQL has been developed based on SQL92. This document describes the subset of the SQL grammar supported by ADQL. Special restrictions and extensions to

SQL92 have been defined in order to support generic and astronomy specific operations.

## Status of This Document

This is a Working Draft. The first release of this document was 2006 January 22.

*This is an IVOA Working Draft for review by IVOA members and other interested parties. It is a draft document and may be updated, replaced, or obsoleted by other documents at any time. It is inappropriate to use IVOA Working Drafts as reference materials or to cite them as other than “work in progress”.*

A list of [current IVOA Recommendations and other technical documents](http://www.ivoa.net/Documents/) can be found at <http://www.ivoa.net/Documents/>.

## Contents

1	Introduction	2
2	Astronomical Data Query Language (ADQL)	3
2.1	Characters, Keywords, Identifiers and Literals	3
2.1.1	Characters	3
2.1.2	Keywords and Identifiers	4
2.1.3	Literals	6
2.2	Query syntax	6
2.2.1	Table subqueries and Joins	7
2.2.2	Search condition	7
2.3	Functions	7
2.3.1	Mathematical Functions	8
2.3.2	Region	8
2.3.3	User Defined Functions	9
	Appendix A: BNF Grammar	10
	References	20

## 1 Introduction

The Astronomical Data Query Language (ADQL) is the language used by the International Virtual Observatory Alliance (IVOA) to represent astronomy queries posted to VO services. The IVOA has developed several standardized protocols to access astronomical data, e.g., SIAP and SSAP for image and spectral data respectively. These protocols might be satisfied using a single table query. However, different VO services have different needs in terms of query complexity and ADQL arises in this context.

The ADQL specification pretends to avoid any distinction between core and advanced or extended functionalities. Hence ADQL has been built according to a single language definition (BNF based [1]). Any service making use of ADQL would then define the level of compliancy to the language. This would allow the notion of core and extension to be service-driven and it would decouple the language from the service specifications.

ADQL is based on the Structured Query Language (SQL), especially on SQL 92. The VO has a number of tabular data sets and many of them are stored in relational databases, making SQL a convenient access means. A subset of the SQL grammar has been extended to support queries which are specific to astronomy.

## 2 Astronomical Data Query Language (ADQL)

This section describes the ADQL language specification. We will define in subsequent sections the syntax for the special characters, reserved and non-reserved words, identifiers and literals and then, finally, the syntax for the query expression.

The formal notation for syntax of computing languages is often expressed in the “Backus Naur Form” BNF. This syntax is used by popular tools for producing parsers. Appendix A to this document provides the full BNF grammar for ADQL.

The following conventions are used through this document:

- Optional items are enclosed in meta symbols [ and ]
- A group of items is enclosed in meta symbols { and }
- Repetitive item (zero or more times) are followed by [...]
- Terminal symbols are enclosed by < and >
- Terminals of meta-symbol characters (=, [, ], (, ), <, >, \*) are surrounded by quotes (“”) to distinguish them from meta-symbols
- Case insensitiveness otherwise stated.

### 2.1 Characters, Keywords, Identifiers and Literals

#### 2.1.1 Characters

The language allows simple Latin letters (lower and upper case, i.e. {aA-zZ}), digits ({0-9}) and the following special characters:

- space
- single quote (')

- double quote (“)
- percent (%)
- left and right parenthesis
- asterisk (\*)
- plus sign (+)
- minus sign (-)
- comma (,)
- period (.)
- solidus (/)
- colon (:)
- semicolon (;)
- less than operator (<)
- equals operator (=)
- greater than operator (>)
- underscore (\_)
- ampersand (&)
- question mark (?)
- vertical bar (|)

### 2.1.2 Keywords and Identifiers

Besides the character set, the language provides a list of reserved keywords plus the syntax description for regular identifiers.

A reserved keyword has a special meaning in ADQL and cannot be used as an identifier. These keywords must be enforced and should be extensive as an escaping mechanism is already in place. We can extend the list of SQL92 reserved keywords to accommodate those useful for astronomical purposes and/or present in a subset of vendor specific languages only (e.g. TOP). This leads to the following list:

- SQL reserved keywords:

ABSOLUTE, ACTION, ADD, ALL, ALLOCATE, ALTER, AND, ANY, ARE, AS, ASC, ASSERTION, AT, AUTHORIZATION, AVG, BEGIN, BETWEEN, BIT, BIT\_LENGTH, BOTH, BY, CASCADE, CASCADED, CASE, CAST, CATALOG, CHAR, CHARACTER, CHARACTER\_LENGTH, CHAR\_LENGTH, CHECK, CLOSE, COALESCE, COLLATE, COLLATION, COLUMN, COMMIT, CONNECT, CONNECTION, CONSTRAINT, CONSTRAINTS, CONTINUE, CONVERT, CORRESPONDING, COUNT, CREATE, CROSS, CURRENT, CURRENT\_DATE, CURRENT\_TIME, CURRENT\_TIMESTAMP, CURRENT\_USER, CURSOR, DATE, DAY, DEALLOCATE, DECIMAL, DECLARE, DEFAULT, DEFERRABLE, DEFERRED, DELETE, DESC, DESCRIBE, DESCRIPTOR, DIAGNOSTICS, DISCONNECT, DISTINCT,

DOMAIN, DOUBLE, DROP, ELSE, END, END-EXEC, ESCAPE, EXCEPT, EXCEPTION, EXEC, EXECUTE, EXISTS, EXTERNAL, EXTRACT, FALSE, FETCH, FIRST, FLOAT, FOR, FOREIGN, FOUND, FROM, FULL, GET, GLOBAL, GO, GOTO, GRANT, GROUP, HAVING, HOUR, IDENTITY, IMMEDIATE, IN, INDICATOR, INITIALLY, INNER, INPUT, INSENSITIVE, INSERT, INT, INTEGER, INTERSECT, INTERVAL, INTO, IS, ISOLATION, JOIN, KEY, LANGUAGE, LAST, LEADING, LEFT, LEVEL, LIKE, LOCAL, LOWER, MATCH, MAX, MIN, MINUTE, MODULE, MONTH, NAMES, NATIONAL, NATURAL, NCHAR, NEXT, NO, NOT, NULL, NULLIF, NUMERIC, OCTET\_LENGTH, OF, ON, ONLY, OPEN, OPTION, OR, ORDER, OUTER, OUTPUT, OVERLAPS, PAD, PARTIAL, POSITION, PRECISION, PREPARE, PRESERVE, PRIMARY, PRIOR, PRIVILEGES, PROCEDURE, PUBLIC, READ, REAL, REFERENCES, RELATIVE, RESTRICT, REVOKE, RIGHT, ROLLBACK, ROWS, SCHEMA, SCROLL, SECOND, SECTION, SELECT, SESSION, SESSION\_USER, SET, SIZE, SMALLINT, SOME, SPACE, SQL, SQLCODE, SQLERROR, SQLSTATE, SUBSTRING, SUM, SYSTEM\_USER, TABLE, TEMPORARY, THEN, TIME, TIMESTAMP, TIMEZONE\_HOUR, TIMEZONE\_MINUTE, TO, TRAILING, TRANSACTION, TRANSLATE, TRANSLATION, TRIM, TRUE, UNION, UNIQUE, UNKNOWN, UPDATE, UPPER, USAGE, USER, USING, VALUE, VALUES, VARCHAR, VARYING, VIEW, WHEN, WHENEVER, WHERE, WITH, WORK, WRITE, YEAR, ZONE

- ADQL reserved keywords:

ABS, ACOS, ASIN, ATAN, ATAN2, CEILING, COS, DEGREES, EXP, FLOOR, LOG, LOG10, MODE, PI, POWER, RADIANS, REGION, RAND, ROUND, SIN, SQRT, SQUARE, TAN, TOP, TRUNCATE, UDF.

The identifiers are used to express, for example, a table or a column reference name.

Both the identifiers and the keywords are case insensitive. They SHALL begin with a letter {aA-zZ}. Subsequent characters shall be letters, underscores or digits {0-9} as follows:

<Latin letter> [{ <underscore> | {<Latin letter> | <digit>} ]

For practical purposes the language specification should be able to address reserved keyword and special character conflicts. To do so the language provides a way to escape a non-compliant identifier by using the double quote character as a delimiter.

ADQL allows making use of the same quoting mechanism to handle the case sensitiveness if needed.

### 2.1.3 Literals

Finally we define the syntax rules for the different data types: string and number.

A string literal is a character expression delimited by single quotes.

Literal numbers are expressed in BNF as follows:

```
<unsigned numeric literal> ::=
    <exact numeric literal> | <approximate numeric literal>
<exact numeric literal> ::=
    <unsigned integer> [<period> [<unsigned integer>]]
    | <period><unsigned integer>
<unsigned integer> ::= <digit>
<approximate numeric literal> ::= <mantissa> E <exponent>
<mantissa> ::= <exact numeric literal>
<exponent> ::= <signed integer>
<signed integer> ::= [<sign>] <unsigned integer>
<sign> ::= <plus sign> | <minus sign>
```

## 2.2 Query syntax

The compact syntax for the SELECT statement shows the main constructs for the query specification:

```
SELECT [ ALL | DISTINCT ]
    [ TOP <unsigned integer> ]
    { * | COUNT(*) | { <expression> [ AS <name> ] } [,...] }
    [ INTO <target specification> ]
    FROM <table reference> [ <alias> ] [,...]
    [ { [NATURAL] { { LEFT | RIGHT } [OUTER] } | INNER | FULL
        [OUTER] } JOIN <alias> { ON <search condition> | USING(col1,
        col2,...) } ]
    [ WHERE <search condition> ]
    [ GROUP BY <column> [,...] ]
    [ HAVING <search condition> ]
    [ ORDER BY { <column> | <unsigned integer> } [ ASC | DESC ]
    [,...]]
```

The SELECT statement defines a query to some derived table(s) specified in the FROM clause. As a result of this query, a subset of the table(s) is returned. The order of the rows MAY be arbitrary unless ORDER BY clause is specified. The

order of the columns to return SHOULD be the same as the order specified in the selection list, or the order defined in the original table if asterisk is specified.

TOP n construct is used to return the first n-rows.

The selection list MAY include any numeric, string value expression.

In the following sections some constructs requiring further description are presented.

### **2.2.1 Table subqueries and Joins**

Table subqueries are present and can be used by some existing predicates within the search condition (IN and BETWEEN most likely) or as an artifact of building derived tables..

Among the different types of joins ADQL supports qualified ones only. These are INNER and OUTER ones (LEFT, RIGHT and FULL). All of these can be NATURAL or not. The join condition does not support embedded sub joins.

### **2.2.2 Search condition**

The search condition can be part of several other clauses: JOIN, HAVING and, obviously, WHERE. Standard logical operators are present in its description (AND, OR and NOT). Six different types of predicates are present in which different types of reserved keywords or characters are used:

- Standard comparison operators: =, !=, <>, <, >, <=, >=
- BETWEEN
- LIKE
- NULL
- EXISTS

[The full syntax for the SELECT statement is present in Appendix A.]

## **2.3 Functions**

ADQL declares a list of non reserved keywords (section 2.1.2) which defines a set of special functions to enhance the astronomical usage of the language. These can be split into three different types:

### 2.3.1 Mathematical Functions

The next table shows the description of the mathematical functions.

Name	Return type	Comment
acos(x)	double	Basic function. Inverse cosine.
asin(x)	double	Basic function. Inverse sine.
atan(x)	double	Basic function. Inverse tangent.
atan2(x,y)	double	Basic function. Inverse tangent of x/y.
cos(x)	double	Basic function. Cosine.
sin(x)	double	Basic function. Sine.
tan(x)	double	Basic function. Tangent.
abs(x)	double	Basic function. Absolute value.
ceiling(x)	double	Basic function. Smallest integer not less than argument.
degrees(x)	double	Basic function. Radians to degrees.
exp(x)	double	Basic function. Exponential.
floor(x)	double	Basic function. Larger integer not greater than argument.
log(x)	double	Basic function. Natural logarithm.
log10(x)	double	Basic function. Base 10 logarithm.
mod(x, y)	double	Basic function. Remainder of y/x.
pi()	double	Basic function. Pi constant.
power(x, y)	double	Basic function. X raised to the power of Y.
radians(x)	double	Basic function. Degree to radians.
sqrt(x)	double	Basic function. Square root.
rand(x)	double	Basic function. Random value between 0.0 and 1.0. It can take a seed value.
round(x, n)	double	Basic function. Round to nearest integer.
truncate(x, n)	double	Basic function. Truncate to n decimal places.

### 2.3.2 Region

To be discussed.



### **2.3.3 User Defined Functions**

To be discussed.

## Appendix A: BNF Grammar

```
<ADQL_language_character> ::=  
  <simple_Latin_letter>  
  | <digit>  
  | <SQL_special_character>
```

```
<ADQL_reserved_word> ::=  
  ABS  
  | ACOS  
  | ASIN  
  | ATAN  
  | ATAN2  
  | CEILING  
  | COS  
  | DEGREES  
  | EXP  
  | FLOOR  
  | LOG  
  | LOG10  
  | MODE  
  | PI  
  | POWER  
  | RADIANS  
  | REGION  
  | RAND  
  | ROUND  
  | SIN  
  | SQRT  
  | SQUARE  
  | TAN  
  | TOP  
  | TRUNCATE  
  | UDF
```

```
<SQL_embedded_language_character> ::=  
  <left_bracket>  
  | <right_bracket>
```

```
<SQL_reserved_word> ::=  
  ABSOLUTE | ACTION | ADD | ALL | ALLOCATE | ALTER | AND | ANY | ARE  
  | AS | ASC | ASSERTION | AT | AUTHORIZATION | AVG  
  | BEGIN | BETWEEN | BIT | BIT_LENGTH | BOTH | BY  
  | CASCADE | CASCADED | CASE | CAST | CATALOG | CHAR | CHARACTER  
  | CHARACTER_LENGTH  
  | CHAR_LENGTH | CHECK | CLOSE | COALESCE | COLLATE | COLLATION  
  | COLUMN | COMMIT  
  | CONNECT | CONNECTION | CONSTRAINT | CONSTRAINTS | CONTINUE  
  | CONVERT | CORRESPONDING | COUNT  
  | CREATE | CROSS | CURRENT | CURRENT_DATE | CURRENT_TIME  
  | CURRENT_TIMESTAMP  
  | CURRENT_USER | CURSOR
```

| DATE | DAY | DEALLOCATE | DECIMAL | DECLARE | DEFAULT  
| DEFERRABLE | DEFERRED | DELETE | DESC | DESCRIBE | DESCRIPTOR  
| DIAGNOSTICS  
| DISCONNECT | DISTINCT | DOMAIN | DOUBLE | DROP  
| ELSE | END | END-EXEC | ESCAPE | EXCEPT | EXCEPTION | EXEC  
| EXECUTE  
| EXISTS | EXTERNAL | EXTRACT  
| FALSE | FETCH | FIRST | FLOAT | FOR | FOREIGN | FOUND | FROM | FULL  
| GET | GLOBAL | GO | GOTO | GRANT | GROUP  
| HAVING | HOUR  
| IDENTITY | IMMEDIATE | IN | INDICATOR | INITIALLY | INNER | INPUT  
| INSENSITIVE  
| INSERT | INT | INTEGER | INTERSECT | INTERVAL | INTO | IS  
| ISOLATION  
| JOIN  
| KEY  
| LANGUAGE | LAST | LEADING | LEFT | LEVEL | LIKE | LOCAL | LOWER  
| MATCH | MAX | MIN | MINUTE | MODULE | MONTH  
| NAMES | NATIONAL | NATURAL | NCHAR | NEXT | NO | NOT | NULL  
| NULLIF | NUMERIC  
| OCTET\_LENGTH | OF | ON | ONLY | OPEN | OPTION | OR | ORDER | OUTER  
| OUTPUT | OVERLAPS  
| PAD | PARTIAL | POSITION | PRECISION | PREPARE | PRESERVE | PRIMARY  
| PRIOR | PRIVILEGES | PROCEDURE | PUBLIC  
| READ | REAL | REFERENCES | RELATIVE | RESTRICT | REVOKE | RIGHT  
| ROLLBACK | ROWS  
| SCHEMA | SCROLL | SECOND | SECTION | SELECT | SESSION  
| SESSION\_USER | SET  
| SIZE | SMALLINT | SOME | SPACE | SQL | SQLCODE | SQLERROR  
| SQLSTATE | SUBSTRING | SUM | SYSTEM\_USER  
| TABLE | TEMPORARY | THEN | TIME | TIMESTAMP | TIMEZONE\_HOUR  
| TIMEZONE\_MINUTE  
| TO | TRAILING | TRANSACTION | TRANSLATE | TRANSLATION | TRIM | TRUE  
| UNION | UNIQUE | UNKNOWN | UPDATE | UPPER | USAGE | USER | USING  
| VALUE | VALUES | VARCHAR | VARYING | VIEW  
| WHEN | WHENEVER | WHERE | WITH | WORK | WRITE  
| YEAR  
| ZONE

<SQL\_special\_character> ::=

<ampersand>  
| <asterisk>  
| <colon>  
| <comma>  
| <double\_quote>  
| <equals\_operator>  
| <greater\_than\_operator>  
| <left\_paren>  
| <less\_than\_operator>  
| <minus\_sign>  
| <percent>  
| <period>  
| <plus\_sign>  
| <question\_mark>  
| <quote>  
| <right\_paren>

| <semicolon>  
| <solidus>  
| <space>  
| <underscore>  
| <vertical\_bar>

<ampersand> ::= &

<approximate\_numeric\_literal> ::= <mantissa> E <exponent>

<as\_clause> ::= [ AS ] <column\_name>

<asterisk> ::= \*

<between\_predicate> ::=  
 <value\_expression> [ NOT ] BETWEEN <value\_expression> AND <value\_expression>

<boolean\_factor> ::= [ NOT ] <boolean\_primary>

<boolean\_primary> ::=  
 <predicate> | <left\_paren> <search\_condition> <right\_paren>

<boolean\_term> ::= <boolean\_factor> | <boolean\_term> AND <boolean\_factor>

<catalog\_name> ::= <identifier>

<character\_factor> ::= <character\_primary>

<character\_primary> ::= <value\_expression\_primary> | <user\_defined\_function>

<character\_representation> ::= <nonquote\_character> | <quote\_symbol>

<character\_string\_literal> ::=  
 <quote> [ <character\_representation>... ] <quote>  
 [ {<separator>... <quote> [ <character\_representation>... ]<quote>}... ]

<character\_value\_expression> ::= <concatenation> | <character\_factor>

<colon> ::= :

<column\_name> ::= <identifier>

<column\_name\_list> ::= <column\_name> [ { <comma> <column\_name> }... ]

<column\_reference> ::= [ <qualifier> <period> ] <column\_name>

<comma> ::= ,

<comment> ::= <comment\_introducer> [ <comment\_character>... ] <newline>

<comment\_character> ::= <nonquote\_character> | <quote>

<comment\_introducer> ::= <minus\_sign><minus\_sign> [ <minus\_sign>... ]

<comp\_operator> ::=

<equals\_operator>  
 | <not\_equals\_operator>  
 | <less\_than\_operator>  
 | <greater\_than\_operator>  
 | <less\_than\_or\_equals\_operator>  
 | <greater\_than\_or\_equals\_operator>

<comparison\_predicate> ::= <value\_expression> <comp\_operator> <value\_expression>

<concatenation> ::=  
 <character\_value\_expression> <concatenation\_operator> <character\_factor>

<concatenation\_operator> ::= ||

<correlation\_name> ::= <identifier>

<correlation\_specification> ::= [ AS ] <correlation\_name>

<default\_function\_prefix> ::=  
 !! The prefix is set by default to "udf."  
 !! It should be possible to change the default prefix to accommodate local preferences

<delimited\_identifier> ::= <double\_quote> <delimited\_identifier\_body> <double\_quote>

<delimited\_identifier\_body> ::= <delimited\_identifier\_part>...

<delimited\_identifier\_part> ::= <nondoublequote\_character> | <doublequote\_symbol>

<delimiter\_token> ::=  
 <character\_string\_literal>  
 | <delimited\_identifier>  
 | <ADQL\_special\_character>  
 | <SQL\_special\_character>  
 | <not\_equals\_operator>  
 | <greater\_than\_or\_equals\_operator>  
 | <less\_than\_or\_equals\_operator>  
 | <concatenation\_operator>  
 | <double\_period>  
 | <left\_bracket>  
 | <right\_bracket>

<derived\_column> ::= <value\_expression> [ <as\_clause> ]

<derived\_table> ::= <table\_subquery>

<digit> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

<double\_period> ::= ..

<double\_quote> ::= "

<double\_quote\_symbol> ::= <double\_quote><double\_quote>

<equals\_operator> ::= =

<exact\_numeric\_literal> ::=

<unsigned\_integer> [ <period> [ <unsigned\_integer> ] ]  
 | <period> <unsigned\_integer>

<exists\_predicate> ::= EXISTS <table\_subquery>

<exponent> ::= <signed\_integer>

<factor> ::= [ <sign> ] <numeric\_primary>

<from\_clause> ::= FROM <table\_reference> [ { <comma> <table\_reference> }... ]

<general\_literal> ::= <character\_string\_literal>

<general\_set\_function> ::=  
 <set\_function\_type> <left\_paren> [ <set\_quantifier> ] <value\_expression> <right\_paren>

<greater\_than\_operator> ::= >

<greater\_than\_or\_equals\_operator> ::= >=

<group\_by\_clause> ::= GROUP BY <grouping\_column\_reference\_list>

<grouping\_column\_reference> ::= <column\_reference>

<grouping\_column\_reference\_list> ::=  
 <grouping\_column\_reference> [ { <comma> <grouping\_column\_reference> }... ]

<having\_clause> ::= HAVING <search\_condition>

<identifier> ::= <regular\_identifier> | <delimited\_identifier>

<in\_predicate> ::= <value\_expression> [ NOT ] IN <in\_predicate\_value>

<in\_predicate\_value> ::=  
 <table\_subquery> | <left\_paren> <in\_value\_list> <right\_paren>

<in\_value\_list> ::=  
 <value\_expression> { <comma> <value\_expression> } ...

<join\_column\_list> ::= <column\_name\_list>

<join\_condition> ::= ON <search\_condition>

<join\_specification> ::= <join\_condition> | <named\_columns\_join>

<join\_type> ::=  
 INNER  
 | <outer\_join\_type> [ OUTER ]

<joined\_table> ::=  
 <qualified\_join>  
 | <left\_paren> <joined\_table> <right\_paren>

<keyword> ::= <SQL\_reserved\_word> | <ADQL\_reserved\_word>

<left\_bracket> ::= [

<left\_paren> ::= (  
 <less\_than\_operator> ::= <  
 <less\_than\_or\_equals\_operator> ::= <=  
 <like\_predicate> ::=  
     <match\_value> [ NOT ] LIKE <pattern>  
 <mantissa> ::= <exact\_numeric\_literal>  
 <match\_value> ::= <character\_value\_expression>  
 <math\_function> ::=  
     ABS <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | CEILING <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | DEGREES <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | EXP <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | FLOOR <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | LOG <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | PI <left\_paren><right\_paren>  
     | POWER <left\_paren> <numeric\_value\_expression> <comma> <unsigned\_integer>  
     <right\_paren>  
     | RADIANS <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | SQUARE <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | SQRT <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | LOG10 <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | RAND <left\_paren> [ <numeric\_value\_expression> ] <right\_paren>  
     | ROUND <left\_paren> <numeric\_value\_expression> <right\_paren>  
     | TRUNCATE <left\_paren> <numeric\_value\_expression> <right\_paren>  
 <minus\_sign> ::= -  
 <named\_columns\_join> ::= USING <left\_paren> <join\_column\_list> <right\_paren>  
 <newline> ::= !! implementation defined end of line indicator  
 <nondelimiter\_token> ::=  
     <regular\_identifier>  
     | <keyword>  
     | <unsigned\_numeric\_literal>  
 <nondoublequote\_character> ::= !! See syntax rules  
 <nonquote\_character> ::= !! One ASCII character  
 <not\_equals\_operator> ::= <not\_equals\_operator1> | <not\_equals\_operator2>  
 <not\_equals\_operator1> ::= <>  
 <not\_equals\_operator2> ::= !=  
 <>null\_predicate> ::= <column\_reference> IS [ NOT ] NULL  
 <numeric\_primary> ::= <value\_expression\_primary> | <numeric\_value\_function>

<numeric\_value\_expression> ::=  
     <term>  
     | <numeric\_value\_expression> <plus\_sign> <term>  
     | <numeric\_value\_expression> <minus\_sign> <term>

<numeric\_value\_function> ::=  
     <trigonometric\_function>  
     | <math\_function>  
     | <user\_defined\_function>

<order\_by\_clause> ::= ORDER BY <sort\_specification\_list>

<ordering\_specification> ::= ASC | DESC

<outer\_join\_type> ::= LEFT | RIGHT | FULL

<pattern> ::= <character\_value\_expression>

<percent> ::= %

<period> ::= .

<plus\_sign> ::= +

<predicate> ::=  
     <comparison\_predicate>  
     | <between\_predicate>  
     | <in\_predicate>  
     | <like\_predicate>  
     | <null\_predicate>  
     | <exists\_predicate>  
     | <region\_predicate>

<qualified\_join> ::=  
     <table\_reference> [ NATURAL ] [ <join\_type> ] JOIN <table\_reference> [ <join\_specification> ]

<qualifier> ::= <table\_name> | <correlation\_name>

<query\_expression> ::=  
     <query\_specification>  
     | <joined\_table>

<query\_specification> ::=  
     SELECT [ <set\_quantifier> ] [ <set\_limit> ] <select\_list> <table\_expression>

<question\_mark> ::= ?

<quote> ::= '

<quote\_symbol> ::= <quote> <quote>

<region\_predicate> ::= !! To be defined

<region\_specification> ::= !! Region definition from STC schema here



```

<regular_identifier> ::=
  <simple_Latin_letter>... [ { <underscore> | <simple_Latin_letter> | <digit> } ... ]

<right_bracket> ::= ]

<right_paren> ::= )

<schema_name> ::= [ <catalog_name> <period> ] <unqualified_schema_name>

<search_condition> ::=
  <boolean_term>
  | <search_condition> OR <boolean_term>

<select_list> ::=
  <asterisk>
  | <select_sublist> [ { <comma> <select_sublist> }... ]

<select_sublist> ::= <derived_column> | <qualifier> <period> <asterisk>

<semicolon> ::= ;

<separator> ::= { <comment> | <space> | <newline> }...

<set_function_specification> ::=
  COUNT <left_paren> <asterisk> <right_paren>
  | <general_set_function>

<set_function_type> ::= AVG | MAX | MIN | SUM | COUNT

<set_limit> ::= TOP <unsigned_integer>

<set_quantifier> ::= DISTINCT | ALL

<sign> ::= <plus_sign> | <minus_sign>

<signed_integer> ::= [ <sign> ] <unsigned_integer>

<simple_Latin_letter> ::=
  <simple_Latin_upper_case_letter>
  | <simple_Latin_lower_case_letter>

<simple_Latin_lower_case_letter> ::=
  a | b | c | d | e | f | g | h | i | j | k | l | m | n
  | o | p | q | r | s | t | u | v | w | x | y | z

<simple_Latin_upper_case_letter> ::=
  A | B | C | D | E | F | G | H | I | J | K | L | M | N
  | O | P | Q | R | S | T | U | V | W | X | Y | Z

<solidus> ::= /

<sort_key> ::= <column_name> | <unsigned_integer>

<sort_specification> ::= <sort_key> [ <ordering_specification> ]

<sort_specification_list> ::= <sort_specification> [ { <comma> <sort_specification> }... ]

```

<space> ::= !! space character here

<string\_value\_expression> ::= <character\_value\_expression>

<subquery> ::=  
 <left\_paren> <query\_expression> <right\_paren>

<table\_expression> ::=  
 <from\_clause>  
 [ <where\_clause> ]  
 [ <group\_by\_clause> ]  
 [ <having\_clause> ]  
 [ <order\_by\_clause> ]

<table\_name> ::= [ <schema\_name> <period> ] <identifier>

<table\_reference> ::=  
 <table\_name> [ <correlation\_specification> ]  
 | <derived\_table> <correlation\_specification>  
 | <joined\_table>

<table\_subquery> ::= <subquery>

<term> ::=  
 <factor>  
 | <term> <asterisk> <factor>  
 | <term> <solidus> <factor>

<token> ::=  
 <nondelimiter\_token>  
 | <delimiter\_token>

<trigonometric\_function> ::=  
 ACOS <left\_paren> <numeric\_value\_expression> <right\_paren>  
 | ASIN <left\_paren> <numeric\_value\_expression> <right\_paren>  
 | ATAN <left\_paren> <numeric\_value\_expression> <right\_paren>  
 | ATAN2 <left\_paren> <numeric\_value\_expression> <comma> <numeric\_value\_expression>  
 <right\_paren>  
 | COS <left\_paren> <numeric\_value\_expression> <right\_paren>  
 | COT <left\_paren> <numeric\_value\_expression> <right\_paren>  
 | SIN <left\_paren> <numeric\_value\_expression> <right\_paren>  
 | TAN <left\_paren> <numeric\_value\_expression> <right\_paren>

<underscore> ::= \_

<unqualified\_schema\_name> ::= <identifier>

<unsigned\_integer> ::= <digit> ...

<unsigned\_literal> ::= <unsigned\_numeric\_literal> | <general\_literal>

<unsigned\_numeric\_literal> ::=  
 <exact\_numeric\_literal>  
 | <approximate\_numeric\_literal>

<unsigned\_value\_specification> ::= <unsigned\_literal>

<user\_defined\_function> ::=  
 <user\_defined\_function\_name>  
 <left\_paren>  
 [ <numeric\_value\_expression> [ { <comma> <numeric\_value\_expression> } ... ] ]  
 <right\_paren>

<user\_defined\_function\_name> ::=  
 [ <default\_function\_prefix> ] <regular\_identifier>

<value\_expression> ::=  
 <numeric\_value\_expression>  
 | <string\_value\_expression>

<value\_expression\_primary> ::=  
 <unsigned\_value\_specification>  
 | <column\_reference>  
 | <set\_function\_specification>  
 | <left\_paren> <value\_expression> <right\_paren>

<vertical\_bar> ::= |

<where\_clause> ::= WHERE <search\_condition>

## References

[1] BNF Grammar for ISO/IEC 9075:1992 – Database Language SQL (SQL-92)  
<http://savage.net.au/SQL/sql-92.bnf.html>